In[ ]:=
```
v = v0[y] (x / δ[y]) (1 - x / δ[y]) ^ 2
```
Out[ ]=

$$\frac{x\, \text{v0}[y] \left(1-\frac{x}{\delta[y]}\right)^2}{\delta[y]}$$

In[ ]:=
```
T0 = T00 + γ y
```
Out[ ]=

$$\text{T00} + y\,\gamma$$

In[ ]:=
```
T = T0 + (Tw - T0) (1 - x / δ[y]) ^ 2
```
Out[ ]=

$$\text{T00} + y\,\gamma + (-\text{T00} + \text{Tw} - y\,\gamma) \left(1-\frac{x}{\delta[y]}\right)^2$$

In[ ]:=

$$\text{momInt} = \partial_y \int_0^{\delta[y]} v^2\, dx == -\nu\, (\partial_x v\ /.\ x \to 0) + \int_0^{\delta[y]} g\,\beta\,(T - \text{T0})\, dx$$

Out[ ]=

$$\frac{2}{105}\, \text{v0}[y] \times \delta[y]\, \text{v0}'[y] + \frac{1}{105}\, \text{v0}[y]^2\, \delta'[y] == -\frac{\nu\, \text{v0}[y]}{\delta[y]} - \frac{1}{3}\, g\,\beta\, (\text{T00} - \text{Tw} + y\,\gamma)\, \delta[y]$$

In[ ]:=

$$\text{egyInt} = \partial_y \int_0^{\delta[y]} v\,(T - \text{T0})\, dx == -\alpha\, (\partial_x T\ /.\ x \to 0)$$

Out[ ]=

$$-\frac{1}{30}\,\gamma\, \text{v0}[y] \times \delta[y] - \frac{1}{30}\,(\text{T00} - \text{Tw} + y\,\gamma)\, \delta[y]\, \text{v0}'[y] - \frac{1}{30}\,(\text{T00} - \text{Tw} + y\,\gamma)\, \text{v0}[y]\, \delta'[y] == \frac{2\,\alpha\,(-\text{T00} + \text{Tw} - y\,\gamma)}{\delta[y]}$$

In[ ]:=
```
nonDimRules = {
    y → yt H,
    δ[y] → δt[yt] H RaH ^ (-1 / 4),
    δ'[y] → δt'[yt] RaH ^ (-1 / 4),
    v0[y] → (α / H) RaH ^ (1 / 2) v0t[yt],
    v0'[y] → (α / H ^ 2) RaH ^ (1 / 2) v0t'[yt],
    γ → b (Tw - T00) / H,
    g → RaH ν α / (β (Tw - T00) H ^ 3)};
simplifyRules = {
    Tw - T00 > 0,
    α > 0, H > 0, RaH > 0, Pr > 0};
```

In[ ]:=
```
momNonDim = momInt /. nonDimRules // # /. ν → α Pr & // Simplify[#, simplifyRules] &
```
Out[ ]=

$$35\,\text{Pr}\,(-1 + b\,\text{yt})\, \delta t[\text{yt}] + \text{v0t}[\text{yt}] \left(\frac{105\,\text{Pr}}{\delta t[\text{yt}]} + 2\,\delta t[\text{yt}]\, \text{v0t}'[\text{yt}]\right) + \text{v0t}[\text{yt}]^2\, \delta t'[\text{yt}] == 0$$

In[ ]:=
```
momPrInf = momNonDim // #[[1]] / Pr & // Limit[#, Pr → ∞] & // # /. b → 1 & // # == 0 &
```
Out[ ]=

$$\frac{105\, \text{v0t}[\text{yt}]}{\delta t[\text{yt}]} + 35\,(-1 + \text{yt})\, \delta t[\text{yt}] == 0$$

In[∘]:=
```
v0tSol = Solve[momPrInf, v0t[yt]][[1]][[1]][[2]]
```
Out[∘]=

$$-\frac{1}{3}\,(-1+yt)\,\delta t[yt]^{2}$$

In[∘]:=
```
egyNonDim = egyInt /. nonDimRules // # /. ν → α Pr & // Simplify[#, simplifyRules] &
```
Out[∘]=

$$\frac{(-1+b\,yt)\,\left(-60+\delta t[yt]^{2}\,v0t'[yt]\right)}{\delta t[yt]}+v0t[yt]\,(b\,\delta t[yt]+(-1+b\,yt)\,\delta t'[yt])==0$$

In[∘]:=
```
problemDetails = {b → 1, v0t[yt] → v0tSol, v0t'[yt] → D[v0tSol, yt]}
```
Out[∘]=

$$\left\{b\to 1,\ v0t[yt]\to -\frac{1}{3}\,(-1+yt)\,\delta t[yt]^{2},\ v0t'[yt]\to -\frac{1}{3}\,\delta t[yt]^{2}-\frac{2}{3}\,(-1+yt)\,\delta t[yt]\,\delta t'[yt]\right\}$$

In[∘]:=
```
egyPrInf = egyNonDim // # /. problemDetails & // Simplify[#, {δt[yt] > 0}] &
```
Out[∘]=

$$(-1+yt)\,\left(180+2\,\delta t[yt]^{4}+3\,(-1+yt)\,\delta t[yt]^{3}\,\delta t'[yt]\right)==0$$

In[∘]:=
```
transformationRules = {δt[yt] → Δ[yt]^(1/4), δt'[yt] → Δ'[yt] × Δ[yt]^(-3/4)/4}
```
Out[∘]=

$$\left\{\delta t[yt]\to \Delta[yt]^{1/4},\ \delta t'[yt]\to \frac{\Delta'[yt]}{4\,\Delta[yt]^{3/4}}\right\}$$

In[∘]:=
```
ΔP = egyPrInf /. transformationRules // FullSimplify[#, {Δ[yt] > 0, yt < 1}] &
```
Out[∘]=

$$720+8\,\Delta[yt]+3\,(-1+yt)\,\Delta'[yt]==0$$

In[∘]:=
```
Solve[ΔP, Δ'[yt]]
```
Out[∘]=

$$\left\{\left\{\Delta'[yt]\to -\frac{8\,(90+\Delta[yt])}{3\,(-1+yt)}\right\}\right\}$$

In[∘]:=
```
sol = NDSolve[{ΔP, Δ[0] == 0}, Δ[yt], {yt, 0, 0.99}]
```
Out[∘]=

{{Δ[yt] → InterpolatingFunction[Domain: {{0., 0.99}} Output: scalar][yt]}}

In[ ]:=
```
Plot[Evaluate[Δ[yt] ^ (1 / 4) /. sol], {yt, 0, 0.99}]
```

Out[ ]=



In[ ]:=
```
hy = ((-κ D[T, x] /. x → 0) / (Tw - T00)) // # /. nonDimRules & //
    # /. transformationRules & // Simplify[#, Tw > T00] & // # /. problemDetails &
```

Out[ ]=

$$-\frac{2\,\mathrm{RaH}^{1/4}\,(-1+\mathrm{yt})\,\kappa}{\mathrm{H}\,\triangle[\mathrm{yt}]^{1/4}}$$

In[ ]:=
```
hyNonDim = hy H / (κ RaH ^ (1 / 4))
```

Out[ ]=

$$-\frac{2\,(-1+\mathrm{yt})}{\triangle[\mathrm{yt}]^{1/4}}$$

In[ ]:=
```
hMeanNonDim = NIntegrate[hyNonDim /. sol, {yt, 0, 1}]
```

Out[ ]=

{0.324668}

In[ ]:=
```
NuMean = hMeanNonDim[[1]] RaH ^ (1 / 4)
```

Out[ ]=

$0.324668\,\mathrm{RaH}^{1/4}$

In[ ]:=
```
NuLiterature = 0.337 RaH ^ (1 / 4)
```

Out[ ]=

$0.337\,\mathrm{RaH}^{1/4}$

In[ ]:=
```
deviation = (1 - NuMean / NuLiterature)
```

Out[ ]=

0.0365936